

CL6 & plus/4 VIC-20 TI-99/4A sinclair

sinclair CL6 & plus/4 VIC-20 TI-99/4A

TI-99/4A sinclair CL6 & plus/4 VIC-20

VIC-20 TI-99/4A sinclair CL6 & plus/4

CL6 & plus/4 VIC-20 TI-99/4A sinclair

sinclair CL6 & plus/4 VIC-20 TI-99/4A

TI-99/4A sinclair CL6 & plus/4 VIC-20

VIC-20 TI-99/4A sinclair CL6 & plus/4

CL6 & plus/4 VIC-20 TI-99/4A sinclair

sinclair CL6 & plus/4 VIC-20 TI-99/4A

TI-99/4A sinclair CL6 & plus/4 VIC-20

VIC-20 TI-99/4A sinclair CL6 & plus/4

CL6 & plus/4 VIC-20 TI-99/4A sinclair

sinclair CL6 & plus/4 VIC-20 TI-99/4A

PAPER soft



44

IL PRIMO SETTIMANALE
DI SOFTWARE SU CARTA PER
CL6 & PLUS/4 - VIC 20 -
TI-99/4A - SINCLAIR

Una pubblicazione della
J.soft editrice
Con la collaborazione del
Gruppo Editoriale Jackson

Anno 2 - N° 44 - 15 novembre 1985

Dots
Mostri galattici
Qui vive
Cinodromo
Astrostorm
Caccia a Rombi
Ma qui è pieno di bug!

L. 1300

VIC-20 TI-99/4A sinclair CL6 & plus/4

speed in dolo post. Cinquop. 11/70

Guida per l'input dei programmi versioni C16 plus/4 - VIC 20

Notate che i listati contengono "parole" racchiuse tra parentesi graffe {}. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da {} deve essere premuto con-

temporaneamente al tasto COMMODE. Inoltre, se tra parentesi graffa si trova un carattere alfabetico "solitario" questo dovrà essere premuto contemporaneamente al tasto CTRL.

Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

{CLR}	SHIFT CLR/HOME		{CYN}	CTRL 4		[<7>]	G 7	
{HOME}	CLR/HOME	S	{PUR}	CTRL 5		[<8>]	G 8	
{SU} "	SHIFT CSR		{GRN}	CTRL 6		{F1}	F1	
{GIU' }	CSR	Q	{BLU}	CTRL 7		{F2}	F2	
{SIN}	SHIFT == CSR ==		{YEL}	CTRL 8		{F3}	F3	
{DES}	== CSR ==	J	[<1>]	G 1		{F4}	F4	
{RVS}	CTRL 5	R	[<2>]	G 2		{F5}	F5	
{OFF}	CTRL 6		[<3>]	G 3		{F6}	F6	
{BLK}	CTRL 1		[<4>]	G 4		{F7}	F7	
{WHT}	CTRL 2	E	[<5>]	G 5		{F8}	F8	
{RED}	CTRL 3		[<6>]	G 6				

Per ricevere gli arretrati di **PAPER** compilare il seguente tagliando:



Compilare ed inviare in busta chiusa a:

J.soft viale Restelli, 5 - 20124 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome _____

Cognome _____

Indirizzo _____

CAP _____ Città _____ Prov. _____

Desidero ricevere i seguenti numeri arretrati di

PAPER _____ Anno _____

PAPER dedicato a _____ Anno _____

a L. 2.000 cad. per un totale di L. _____ a mezzo

assegno allegato

contanti allegati





PAPER soft

Cl6k-plus/4

6 Dots
di C. Goodich trad. e adatt. di F. Sarcina

9 Mostri galattici
di S. Hardin trad. e adatt. di F. Sarcina

VIC-20

12 Qui vive (8K RAM)
di F. Lemmlin trad. e adatt. di F. Sarcina

17 Clnodromo
di F. Voglioso

ti TI-99/4A

21 Astrotorm (BASICO)
di P. Lear trad. e adatt. di E. Re Garbagnati

24 Caccia ai rombi (BASICO)
di R. Didi trad. e adatt. di E. Re Garbagnati

sinclair Spectrum

27 Ma qui è pieno di bug! (16/48K)
di P. Bailey trad. e adatt. di C. Panzalis

J. soft s.r.l.

**DIREZIONE, REDAZIONI
E AMMINISTRAZIONE**
V.le Restelli, 5
20124 Milano
Tel. 68.88.228-68.37.97

DIRETTORE RESPONSABILE:
Pietro Dell'Orco

COORDINAMENTO TECNICO:
Lucio Bragagnolo
Mauro Cristulib Grizzi

REDAZIONE:
Carlo Panzalis
Franco Sarcina

GRAFICA E IMPAGINAZIONE
Margherita La Noce
Ivana Rossi
Raffaella Toffolatti

FOTOCOPOSIZIONE:

Graphotek
Via Astesani, 16 - Milano
Tel. 64.80.397

CONTABILITÀ:
Giulia Pedrazzini
Flavia Bonati

**AUTORIZZAZIONE ALLA
PUBBLICITÀ:**
Aut. Min. Rich.

STAMPA:

Intergrafica - Pioltello (MI)
Rivista associata
all'Unione stampa
Periodica Italiana



PUBBLICITÀ

Concessionario per l'Italia e l'Estero
J. Advertising s.r.l.
V.le Restelli, 5
20124 MILANO
Tel. (02)

68.82.895-68.80.606-68.87.233
Tlx 316213 REINA I
Concessionario esclusivo per la
DIFFUSIONE in Italia e Estero:
SODIP - Via Zuretti, 25
20125 MILANO

Spedizione in abbonamento
postale Gruppo Il/70
Prezzo della rivista L. 1.300
Numero arretrati L. 2.600

© TUTTI I DIRITTI DI
RIPRODUZIONE O TRADUZIONE
DEGLI ARTICOLI PUBBLICATI
SONO RISERVATI



Questo articolo è rivolto esclusivamente ai "Commodoriani" che usano il VIC 20, il C16 o il Plus/4.

Se non l'hai già fatto, memorizza e salva su nastro o disco la versione di STOP BUG per il tuo computer. Questo programma ti permette, una volta caricato ed attivato con RUN, di digitare i listati di Papersoft senza introdurre errori di battitura.

Come avrai notato, le linee dei nostri programmi terminano con la parola: ".rem" seguita da un numero, detto checksum; **NON COPIARE QUESTA PARTE DELLA LINEA**, che serve solo per tua informazione. Quando STOP BUG è in funzione, infatti, tutte le volte che premi il tasto RETURN dopo aver digitato una linea di programma, appare nell'angolo superiore sinistro dello schermo un numero, che deve coincidere con il checksum da noi pubblicato (sempre che tu non abbia usato comandi abbreviati, come ? per PRINT).

Usando STOP BUG tieni conto del

fatto che gli spazi non influiscono sul checksum, per cui fai particolare attenzione alle istruzioni PRINT dove spesso presenti. Inoltre eventuali inversioni di caratteri non vengono segnalate.

Un altro metodo per utilizzare STOP BUG, soprattutto se sei abituato ad usare le abbreviazioni dei comandi Basic, è controllare i checksum listando il programma e battendo RETURN su ogni linea.

Se intendi sospendere il lavoro di digitazione, puoi salvare su nastro o disco il programma che stai memorizzando senza che STOP BUG interferisca; al momento di riprendere il lavoro ricordati però di caricare ed attivare STOP BUG prima di ricaricare il tuo programma.

Per disabilitare STOP BUG premi RUN/STOP e RESTORE (su VIC 20) oppure RUN/STOP e RESET (su C16 o PLUS/4).

Per riattivarlo batti:

SYS PEEK (56)*256 per il VIC 20;
SYS 16128 per il C16 e il Plus/4.

STOP BUG per C16 e PLUS/4

```
100 POKE55,0:POKE56,63:PRINT"{CLR}ATTENDERE PREGO..."
110 FORI=16128TO16260:READA:CK=CK+A:POKEI,A:NEXT
120 IFCK<>14248THENPRINT"{GIU'}ERRORE NELLE ISTRUZIONI DA
    TA":END
130 SYS16128:PRINT"{CLR}{ 2 GIU'}{RVS}STOP BUG{OFF} ATTIV
    ATO.":NEW
200 DATA173,34,3,201,32,208,1,96,141
210 DATA33,63,173,35,3,141,34,63,169
220 DATA32,141,34,3,169,63,141,35,3
230 DATA169,0,133,6,96,32,232,235,133
240 DATA3,134,4,132,5,8,201,13,240
250 DATA17,201,32,240,5,24,101,6,133
260 DATA6,165,3,166,4,164,5,40,96
270 DATA169,13,32,210,255,165,205,141,251
280 DATA3,206,251,3,169,0,133,207,169
290 DATA19,32,210,255,169,18,32,210,255
300 DATA169,58,32,210,255,166,6,169,0
310 DATA133,6,172,33,63,192,232,208,6
320 DATA32,95,164,76,117,63,32,7,208
330 DATA169,32,32,210,255,32,210,255,173
340 DATA251,3,133,205,76,55,63
```

STOP BUG PER VIC 20

```
100 POKE55,0:POKE56,PEEK(56)-1:R=PEEK(56)*256:PRINT"{CLR}
    ATTENDERE PREGO..."
110 FORI=RTOR+132:READA:CK=CK+A:IFA=999THENA=R/256
120 POKEI,A:NEXT
130 IFCK<>22689THENPRINT"{GIU'}ERRORE NELLE ISTRUZIONI DA
    TA":END
140 SYS(R):PRINT"{CLR}{ 2 GIU'}{RVS}STOP BUG{OFF} ATTIVAT
    O.":NEW
200 DATA173,36,3,201,32,208,1,96,141
210 DATA33,999,173,37,3,141,34,999,169
220 DATA32,141,36,3,169,999,141,37,3
230 DATA169,0,133,254,96,32,87,241,133
240 DATA251,134,252,132,253,8,201,13,240
250 DATA17,201,32,240,5,24,101,254,133
260 DATA254,165,251,166,252,164,253,40,96
270 DATA169,13,32,210,255,165,214,141,251
280 DATA3,206,251,3,169,0,133,216,169
290 DATA19,32,210,255,169,18,32,210,255
300 DATA169,58,32,210,255,166,254,169,0
310 DATA133,254,172,33,999,192,87,208,6
320 DATA32,205,189,76,117,999,32,205,221
330 DATA169,32,32,210,255,32,210,255,173
340 DATA251,3,133,214,76,55,999
```

Dots

Dots è un programma educativo creato per insegnare ai bambini in età prescolare a distinguere numeri e lettere. Essi potranno far comparire un disegno, scelto tra quattro possibili, all'interno di una matrice in cui ogni casellina è indicata dalle sue coordinate (numeri per le righe e lettere per le colonne).

I disegni vengono tracciati congiungendo, per mezzo di linee, le caselle della matrice nelle quali appaiono di volta in volta dei pallini. Ogni pallino aggiunge una linea al disegno soltanto se il bambino batte sulla tastiera le coordinate esatte, e via di seguito finché la figura non è completa.

```

100 REM INTESTAZIONE                                     :rem 18
110 PRINT "{CLR}{BLU}":COLOR0,1:COLOR4,1               :rem 165
120 PRINTSPC(10)"{ 3 Q}{ 3 SPAZI}{ 2 Q}{ 2 SPAZI}{ 5 Q}
   { 2 SPAZI}{ 2 Q}"                                     :rem 40
130 PRINTSPC(10)"Q{ 2 SPAZI}Q{SPAZI}Q{ 2 SPAZI}Q
   { 2 SPAZI}Q{ 2 SPAZI}{SPAZI}Q{ 2 SPAZI}Q"         :rem 244
140 PRINTSPC(10)"Q {SPAZI}Q{SPAZI}Q{ 2 SPAZI}Q{SPAZI}
   {SPAZI}Q{SPAZI} { 2 SPAZI}Q"                       :rem 132
150 PRINTSPC(10)"Q {SPAZI}Q{SPAZI}Q{ 2 SPAZI}Q{SPAZI}
   {SPAZI}Q{SPAZI}{ 2 SPAZI}{ 2 SPAZI}Q"             :rem 133
160 PRINTSPC(10)"Q {SPAZI}Q{SPAZI}Q{ 2 SPAZI}Q{SPAZI}
   {SPAZI}Q{SPAZI}{ 2 SPAZI}Q{ 2 SPAZI}Q"           :rem 87
170 PRINTSPC(10)"{ 3 Q}{SPAZI} {SPAZI}{ 2 Q}{SPAZI}
   { 2 SPAZI}Q{SPAZI}{ 2 SPAZI}{SPAZI}{ 2 Q}"       :rem 233
180 L=3162:C=2738:CD=-1024                             :rem 104
190 A$="Z":GOSUB840                                     :rem 48
200 :                                                  :rem 204
210 REM ISTRUZIONI                                     :rem 151
220 PRINTSPC(14)"{ 2 GIU'}{WHT}ISTRUZIONI:"          :rem 130
230 PRINTSPC(8)"{GIU'}QUANDO APPARE LO SCHEMA,"      :rem 223
240 PRINTSPC(9)"CERCA IL PUNTO TONDO."              :rem 18
250 PRINTSPC(9)"{GIU'}BATTI IL NUMERO DELLA":PRINTSPC(11)
   "RIGA E 'RETURN'."                                  :rem 146
260 PRINTSPC(9)"{GIU'}BATTI LA LETTERA DELLA"        :rem 94
270 PRINTSPC(10)"COLONNA E 'RETURN'."                :rem 173
280 PRINTSPC(5)"{ 2 GIU'}{RVS}PREMI UN TASTO PER COMINCIA
   RE.":POKE239,0                                     :rem 8
290 GETKEYS$                                           :rem 219
300 :                                                  :rem 205
310 REM SCELTA DEL DISEGNO                             :rem 18
320 COLOR0,15,0:PRINT "{CLR}"SPC(8)"{ 5 GIU'}[<3>]CHE COSA
   VUOI DISEGNARE?"                                   :rem 120

```

```

330 PRINTSPC(8)"{BLU}{ 3 GIU'}A - FARFALLA (22 PUNTI)"
:rem 20
340 PRINTSPC(10)"{GIU'}B - FUNGO (12 PUNTI)":PRINTSPC(9)"
{GIU'}C - CAVALLO (20 PUNTI)"
:rem 132
350 PRINTSPC(10)"{GIU'}D - CUORE (10 PUNTI)":POKE239,0
:rem 166
360 GETA$:IFA$<"A"ORA$>"D"THEN360
:rem 97
370 :
:rem 212
380 REM DISEGNO LAVAGNA
:rem 130
390 PRINT"{CLR}[<3>]"SPC(11)"{ 2 GIU'}ABCDEFGHIJKLMNOPQR
{HOME}"
:rem 211
400 FORRH=1TO18:FORT=1TO18:POKEL+T+RH*40,79:POKEC+T+RH*40
,78:NEXT:NEXT
:rem 172
410 PRINTSPC(11)"[<3>]{ 20 GIU'}ABCDEFGHIJKLMNOPQR"
:rem 20
420 PRINT"{HOME}{ 3 GIU'}{BLU}";
:rem 7
430 FORI=1TO18:PRINTSPC(8)RIGHT$(STR$(I),2)SPC(19)"[<G>]"
RIGHT$(STR$(I),2):NEXT
:rem 138
440 :
:rem 210
450 REM CARICAMENTO DATI
:rem 197
460 RESTORE
:rem 190
470 READB$:IFB$<>A$THEN470
:rem 245
480 ONASC(A$)-64GOTO650,690,730,770
:rem 140
490 FORT=1TO500:NEXT
:rem 247
500 READA,B,E$,S:IFA=0THEN830
:rem 181
510 POKEA,81:POKEA+CD,B
:rem 101
520 PRINT"{HOME}{ 39 SPAZI}"
:rem 123
530 PRINT"{BLU}{HOME} (+) NUMERO";:GOSUB930:N$=IN$
:rem 86
540 PRINT"{HOME}"SPC(20)"(↑) LETTERA";:GOSUB930:L$=IN$
:rem 77
550 IFES=N$+L$THEN610
:rem 162
560 PRINT"{HOME}{BLK}{ 16 SPAZI}RIPROVA{ 11 SPAZI}"
:rem 50
570 SOUND1,300,10
:rem 162
580 FORT=1TO400:NEXT:SOUND1,500,10
:rem 56
590 FORT=1TO1500:NEXT:GOTO520
:rem 51
600 :
:rem 208
610 FORT=1TO700:NEXT:FORT=1TO18:POKESQ,160:POKESQ+CD,B:IF
SQ=ATHEN630
:rem 146
620 SQ=SQ+S:NEXT
:rem 20
630 SOUND1,600,10:FORT=1TO400:NEXT:SOUND1,800,10:GOTO490
:rem 138
640 :
:rem 212
650 POKE3290,77:POKE3290+CD,0:POKE3292,78:POKE3292+CD,0
:rem 146
660 POKE3331,160:POKE3331+CD,69
:rem 216
670 SQ=3331:GOTO490
:rem 88
680 :
:rem 216
690 POKE3611,160:POKE3611+CD,68:POKE3612,160:POKE3612+CD,
68
:rem 85

```

```

700 POKE3765,160:POKE3765+CD,69           :rem 233
710 SQ=3765:GOTO490                         :rem 94
720 :                                         :rem 211
730 POKE3484,74:POKE3484+CD,0:POKE3485,75:POKE3485+CD,0
                                           :rem 157
740 POKE3445,85:POKE3445+CD,0              :rem 122
750 POKE3256,160:POKE3256+CD,66:SQ=3256:GOTO490 :rem 220
760 :                                         :rem 215
770 PRINT"{HOME}{ 7 GIU'}"                 :rem 249
780 PRINTSPC(17)"{GRN}PAOLA"               :rem 154
790 PRINTSPC(17)"{ 2 GIU'}[<1>]SONIA"      :rem 45
800 PRINTSPC(16)"{ 2 GIU'}[<3>]TIZIANA"    :rem 207
810 POKE3331,160:POKE3331+CD,66:SQ=3331:GOTO490 :rem 199
820 :                                         :rem 212
830 PRINT"{HOME}{ 10 SPAZI}BUON LAVORO! ANCORA?
     { 4 SPAZI}"                             :rem 154
840 READB$:IFB$<>"Z"THEN840                 :rem 48
850 READF,D:IFF=-1ANDA$="Z"THENRETURN      :rem 59
860 IFF=-1THEN890                           :rem 222
870 SOUND1,F,D:FORT=1TOD*16:NEXT           :rem 15
880 GOTO850                                  :rem 118
890 GETY$:IFY$<>"S"ANDY$<>"N"THEN890        :rem 129
900 IFY$="S"THEN320                          :rem 61
910 :                                         :rem 212
920 PRINT"{CLR}";:END                       :rem 75
930 PRINT"? ";:IN$=""                       :rem 93

940 PRINT"{RVS} {OFF}{SIN}";               :rem 234
950 GETA$:IFA$=""THEN940                    :rem 94
960 ZL=LEN(IN$):IFA$=CHR$(20)ANDZLTHENPRINTA$;:IN$=LEFT$(
     IN$,ZL-1)                               :rem 30
970 IFA$=CHR$(13)ANDZLTHENPRINT" ":RETURN   :rem 26
980 IF(A$<"0"ORA$>"R")OR(A$>"9"ANDA$<"A")ORLEN(IN$)=2THEN
     950                                       :rem 67
990 PRINTA$;:IN$=IN$+A$:GOTO940            :rem 92
1000 :                                       :rem 251
1010 DATAA,3451,69,71,40,3295,66,3M,-39   :rem 245
1020 DATA3297,66,30,1,3379,66,5Q,41,3539,66,9Q,40,3617,66
     ,110,39,3699,71,13Q,41                 :rem 39
1030 DATA3779,71,15Q,40,3818,71,16P,39,3815,71,16M,-1,365
     1,71,12I,-41,3807,71,16E,39           :rem 240
1040 DATA3804,71,16B,-1,3763,71,15A,-41,3683,71,13A,-40,3
     605,71,11C,-39                         :rem 11
1050 DATA3523,66,9A,-41                    :rem 169
1060 DATA3363,66,5A,-40,3285,66,3C,-39,3287,66,3E,1,3451,
     66,7I,41,3731,69,14I,40               :rem 10
1070 DATA0,0,0,0                           :rem 38
1080 :                                       :rem 3
1090 DATAB,3770,69,15H,1,3610,68,11H,-40,3607,68,11E,-1,3
     567,68,10E,-40                         :rem 53
1100 DATA3489,68,8G,-39                   :rem 190

```


1110 DATA3494,68,8L,1,3576,68,10N,41,3616,68,11N,40,3613,
 68,11K,-1,3773,68,15K,40 :rem 94
 1120 DATA3778,69,15P,1,3770,69,15H,-1,0,0,0,0 :rem 253
 1130 : :rem 255
 1140 DATAC,3412,66,6J,39,3407,66,6E,-1,3446,66,7D,39,3646
 ,66,12D,40,3728,66,14F,41 :rem 134
 1150 DATA3729,66,14G,1,3689,66,13G,-40,3648,66,12F,-41,36
 08,66,11F,-40,3612,66,11J,1 :rem 204
 1160 DATA3735,66,14M,41,3736,66,14N,1,3656,66,12N,-40,361
 5,66,11M,-41,3455,66,7M,-40 :rem 238
 1170 DATA3377,66,50,-39,3379,0,5Q,1,3339,0,4Q,-40,3338,0,
 4P,-1,3256,66,2N,-41 :rem 138
 1180 DATA0,0,0,0 :rem 40
 1190 : :rem 5
 1200 DATAD,3208,66,1F,-41,3205,66,1C,-1,3283,66,3A,39,352
 3,66,9A,40,3851,66,17I,41 :rem 97
 1210 DATA3539,66,9Q,-39,3299,66,3Q,-40,3217,66,10,-41,321
 4,66,1L,-1,3331,66,4I,39 :rem 88
 1220 DATA0,0,0,0,Z :rem 169
 1230 : :rem 0
 1240 DATA596,30,685,10,739,20,810,20 :rem 32
 1250 DATA739,20,685,20,596,20,685,20,739,30,685,10
 :rem 229
 1260 DATA596,20 :rem 21
 1270 DATA453,20,596,40 :rem 110
 1280 DATA596,10,-1,0 :rem 252

Mostri galattici

Ecco un gioco per i virtuosi della tastiera. Utilizzando i tasti W, A, D, X per andare rispettivamente in alto, a sinistra, a destra e in basso dovete attraversare con la vostra astronave il quadro spaziale dall'alto verso il basso. La manovra sarebbe semplice se non ci fossero a complicarla dei velocissimi mostri che fanno il possibile per catturarvi ed annientarvi. Ogni volta che

la traversata riuscirà vi troverete di fronte ad un nuovo quadro. Ma attenzione: si tratta di una lotta contro il tempo, in quanto per decidere ogni mossa avete a disposizione pochissimi protosecondi, passati i quali ogni possibilità di salvare la pelle è irrimediabilmente persa! Il gioco termina quando tutte e cinque le astronavi sono distrutte.

10 REM MOSTRI GALATTICI
 20 COLOR0,2,7:COLOR4,7,3

:rem 181
 :rem 129

```

30 PRINT"{CLR}{GIU'}{ 11 SPAZI}{RED}{RVS} MOSTRI GALATTIC
   I {OFF}" :rem 8
40 PRINT"{BLU}{ 4 GIU'}{ 2 SPAZI}PER MUOVERE L'ASTRONAVE
   USA I TASTI" :rem 127
50 PRINT"{GIU'}{ 14 SPAZI}->W :SOPRA" :rem 200
60 PRINT"{GIU'}{ 14 SPAZI}->A :SINISTRA" :rem 155
70 PRINT"{GIU'}{ 14 SPAZI}->D :DESTRA" :rem 245
80 PRINT"{GIU'}{ 14 SPAZI}->X :SOTTO" :rem 224
90 PRINT"{ 3 GIU'}{ 5 SPAZI}PREMI UN TASTO PER CONTINUARE
   " :rem 247
100 GETKEYA$ :rem 191
110 PRINT"{CLR}" :rem 246
120 D$="{ 20 GIU'}" :rem 208
130 VOL8 :rem 189
140 DIMX(12) :rem 123
150 DIMY(12) :rem 125
160 GOTO1010 :rem 146
170 REM SCHERMO :rem 141
180 PRINT"{CLR}" :rem 253
190 FORT=1TO10 :rem 71
200 PRINTSPC(13)"[<1>] { 11 Q}" :rem 122
210 NEXTT :rem 38
220 PRINTSPC(13) " [< 11 +>]" :rem 34
230 PRINT"{HOME}{ 2 GIU'}{ 19 DES}{GRN}X{ 10 GIU'}"
   :rem 98
240 PRINTSPC(13)"PUNTEGGIO:"SC :rem 132
250 X1=6 :rem 147
260 Y1=2 :rem 145
270 FORT=1TO11 :rem 71
280 X(T)=T :rem 40
290 Y(T)=11 :rem 56
300 NEXTT :rem 38
310 KA=KA+.1 :rem 115
320 RETURN :rem 117
330 REM MOSSA DEL GIOCATORE :rem 111
340 KY=PEEK(2038) :rem 187
350 SOUND1,700,8 :rem 121
360 PRINT"{HOME}";TAB(X1+13);LEFT$(D$,Y1);"[<1>]Q"
   :rem 194
370 IFKY=18THENX1=X1+1 :rem 77
380 IFKY=9THENY1=Y1-1 :rem 34
390 IFKY=23THENY1=Y1+1 :rem 77
400 IFKY=10THENX1=X1-1 :rem 65
410 IFX1<1THENX1=1 :rem 64
420 IFY1<1THENY1=1 :rem 67
430 IFX1>11THENX1=11 :rem 166
440 PRINT"{HOME}";TAB(X1+13);LEFT$(D$,Y1);"{GRN}X"
   :rem 101
450 PRINT"{HOME}";TAB(X1+13);LEFT$(D$,Y1);"{GRN}X"
   :rem 102
460 FORT=1TO11 :rem 72

```

470	IFX1<>X(T) THEN510	:rem	239
480	IFY1<>Y(T) THEN510	:rem	242
490	GOSUB950	:rem	187
500	CA=-1	:rem	180
510	NEXTT	:rem	41
520	RETURN	:rem	119
530	REM MOSSA DEL MOSTRO	:rem	184
540	FORT=1TO11	:rem	71
550	PRINT"{HOME}";TAB(X(T)+13);LEFT\$(D\$,Y(T));"[<1>]Q"	:rem	171
560	IFRND(0)<KATHEN830	:rem	33
570	A=INT(RND(0)*4)	:rem	25
580	IFA=0THEN620	:rem	161
590	IFA=1THEN640	:rem	165
600	IFA=2THEN660	:rem	160
610	GOTO680	:rem	110
620	X(T)=X(T)+1	:rem	43
630	GOTO690	:rem	113
640	X(T)=X(T)-1	:rem	47
650	GOTO690	:rem	115
660	Y(T)=Y(T)-1	:rem	51
670	GOTO690	:rem	117
680	Y(T)=Y(T)-1	:rem	53
690	IFX(T)<1THEN740	:rem	98
700	IFX(T)>11THEN760	:rem	143
710	IFY(T)<1THEN780	:rem	96
720	IFY(T)>11THEN800	:rem	141
730	GOTO810	:rem	108
740	X(T)=X(T)+1	:rem	46
750	GOTO890	:rem	118
760	X(T)=X(T)-1	:rem	50
770	GOTO890	:rem	120
780	Y(T)=Y(T)+1	:rem	52
790	GOTO890	:rem	122
800	Y(T)=Y(T)-1	:rem	47
810	PRINT"{HOME}";TAB(X(T)+13);LEFT\$(D\$,Y(T));"{GRN}[<+>]	:rem	28
	"	:rem	116
820	GOTO890	:rem	21
830	X2=X1-X(T)	:rem	25
840	Y2=Y1-Y(T)	:rem	2
850	IFRND(0)<.8THEN880	:rem	195
860	X(T)=X(T)+SGN(X2)	:rem	121
870	GOTO890	:rem	200
880	Y(T)=Y(T)+SGN(Y2)	:rem	135
890	PRINT"{HOME}";TAB(X(T)+13);LEFT\$(D\$,Y(T));"[<1>][<+>]	:rem	44
	"	:rem	73
900	NEXTT	:rem	131
910	FORU=1TO11	:rem	48
920	PRINT"{HOME}";TAB(X(U)+13);LEFT\$(D\$,Y(U));"[<1>][<+>]	:rem	
	"	:rem	
930	NEXTU	:rem	

940 RETURN	:rem 125
950 FORL=1TO10	:rem 67
960 FORM=500TO780+L*10STEP20	:rem 135
970 SOUND1,M,1	:rem 48
980 NEXTM	:rem 45
990 NEXTL	:rem 45
1000 RETURN	:rem 161
1010 REM PROGRAMMA CENTRALE	:rem 154
1020 KA=KA+.1	:rem 162
1030 GOSUB180:REM SCHERMO	:rem 12
1040 GOSUB340:REM MOSSA DEL GIOCATORE	:rem 239
1050 BR=BR+1	:rem 135
1060 IFNOTCATHEN1180	:rem 196
1070 K=K+1:CA=0	:rem 34
1080 SC=SC+AT-BR-5*K	:rem 138
1090 IFK<5THEN1030	:rem 8
1100 PRINT"{CLR}{GIU'} TERMINE DEL GIOCO."	:rem 191
1110 PRINT"{GIU'} IL TUO PUNTEGGIO E' DI:";	:rem 82
1120 PRINTSC	:rem 231
1130 PRINT"{GIU'} VUOI CONTINUARE (S/N)?{GIU'}":POKE239,0	:rem 182
1140 GETKEYA\$:rem 244
1150 IFA\$="S"THENRUN	:rem 179
1160 IFA\$<>"N"THEN1140	:rem 189
1170 END	:rem 160
1180 AT=0:IFY1=11THENBR=0:AT=40:GOTO1080	:rem 154
1190 GOSUB540	:rem 228
1200 GOTO1040	:rem 193

VIC-20

Qui vive

8K RAM

Questo gioco, appassionante e antichissimo, si effettua su una scacchiera di dimensioni 5 x 5. Ogni giocatore dispone di 5 pedine e deve cercare di sistemarle in modo da ottenere una configurazione simmetrica. Nella versione che vi presentiamo il vostro avversario è il VIC.

All'inizio del gioco la scacchiera è vuota e, alternativamente, voi e il

computer dovreste disporre le pedine su di essa impostando le coordinate desiderate (A1, B3, D2, ecc.). Una volta sistemate tutte le pedine comincerete a muoverle cercando di ottenere una configurazione vincente. Dovrete in questo caso digitare prima le coordinate della pedina che volete muovere, quindi le coordinate della posizione d'arrivo desiderata.

10	REM QUI VIVE	:rem	110
20	POKE36879,141	:rem	100
30	DIMSF(41,4),SO(41)	:rem	112
40	DIMD(107,1),DF(15,8)	:rem	153
50	DIMCC(4,1),B(24),C(24)	:rem	242
60	GOTO1810	:rem	105
70	REM CALCOLA I PUNTI DI DF	:rem	42
80	GV=0:P1=0:P2=0:C1=0:C2=0	:rem	250
90	D1=B(DF(NF,0)):D2=B(DF(NF,1))	:rem	197
100	IFD1=1ORD2=1THENP1=1	:rem	166
110	IFD1=10ORD2=10THENC1=10	:rem	42
120	D1=B(DF(NF,2)):D2=B(DF(NF,3))	:rem	243
130	IFD1=1ORD2=1THENP2=1	:rem	170
140	IFD1=10OR D2=10THENC2=10	:rem	46
150	GV=B(DF(NF,4))+B(DF(NF,5))+B(DF(NF,6))	:rem	239
160	P=GV+P1+P2+C1+C2	:rem	89
170	IFGV=30AND(C1=0ORC2=0)THENP=P-10	:rem	134
180	IFGV=3AND(P1=0ORP2=0)THENP=P-1	:rem	65
190	RETURN	:rem	122
200	REM INIZIALIZZA LE FIGURE SINGOLE	:rem	40
210	FORI=0TO41:READX\$:rem	10
220	FORJ=1TO5	:rem	11
230	SF(I,J-1)=ASC(MID\$(X\$,J,1))-65	:rem	55
240	NEXT:NEXT:RETURN	:rem	104
250	REM CONFIGURAZIONI	:rem	162
260	DATAUQMSY,PLHNT,KGCIO	:rem	156
270	DATAAGMIE,FLRNJ,KQWSO	:rem	127
280	DATAAGMQU,BHNRV,CIOSW	:rem	140
290	DATAEIMSY,DHLRX,CGKQW	:rem	147
300	REM PER (X)	:rem	7
310	DATAACGKM,BDHLN,CEIMO	:rem	62
320	DATAFLHPR,GIMQS,HJNRT	:rem	138
330	DATAKMQUW,LNRVX,MOSWY	:rem	214
340	REM PIU' (+)	:rem	12
350	DATABFGHL,CGHIM,DHIJN	:rem	66
360	DATAGKLMQ,HLMNR,IMNOS	:rem	142
370	DATALPQRV,MQRSW,NRSTX	:rem	218
380	REM DIAGONALI (/)	:rem	135
390	DATAAGMSY,EIMQU	:rem	228
400	REM COLONNE (I)	:rem	32
410	DATAAFKPU,BGLQV,CHMRW	:rem	123
420	DATADINSX,EJOTY	:rem	237
430	REM RIGHE (-)	:rem	104
440	DATAABCDE,FGHIJ,KLMNO	:rem	66
450	DATAPQRST,UVWXY	:rem	44
460	REM INIZIALIZZA LE FIGURE DOPPIE	:rem	224
470	FORI=0TO107STEP12:READX\$:rem	228
480	FORJ=0TO11:FORK=0TO1	:rem	236
490	L=2*J+K+1:D(I+J,K)=ASC(MID\$(X\$,L,1))-49	:rem	65
500	NEXT:NEXT:NEXT	:rem	198
510	RETURN	:rem	118

520 DATA"171:1A1C1E1M1O1P2>2@2B2J"	:rem	117
530 DATA"393<3=3?3G474:4=4?4A4O4O"	:rem	101
540 DATA"4P5@5B5D5J696<6C6E6M7=7A"	:rem	163
550 DATA"7C7I7O7P8>8B8D8J9?9E9K:?"	:rem	224
560 DATA":A:E:K:O:P;>;@;D;J<=<C<I"	:rem	0
570 DATA"=G=I=O>F>H>J?G?K?P@F@J@L"	:rem	93
580 DATA"AGAIKAMAOPBBHJBNCICMCP"	:rem	157
590 DATA"DJDLDNEKEMEOFRFWGSWHTHW"	:rem	10
600 DATA"IRIXJSJXKTKXLRLYMSMYNTNY"	:rem	131
610 REM VALUTA IL CAMPO DELLE FIGURE DOPPIE	:rem	54
620 TC=4:TN=0	:rem	233
630 FORI=0TO4:C(SF(D(DF,0),I))=1:NEXT	:rem	156
640 FORI=0TO4:SF=SF(D(DF,1),I)	:rem	249
650 IFC(SF)=1THEN670	:rem	145
660 DF(NF, TN)=SF:TN=TN+1:GOTO680	:rem	215
670 DF(NF, TC)=SF:TC=TC+1:C(SF)=0	:rem	122
680 NEXT	:rem	221
690 FORI=0TO4:SF=SF(D(DF,0),I)	:rem	253
700 IFC(SF)=1THENDF(NF, TN)=SF:TN=TN+1:C(SF)=0	:rem	238
710 NEXT:RETURN	:rem	241
720 REM VISUALIZZA IL CAMPO	:rem	148
730 PRINTSPC(5)"{GIU'}{ 3 SPAZI}ABCDE{GIU'}"	:rem	72
740 FORI=0TO4:PRINTSPC(5)5-I;:FORJ=0TO4	:rem	215
750 F=I*5+J	:rem	60
760 IFB(F)=0THENPRINT".";	:rem	219
770 IFB(F)=1THENPRINT"W";	:rem	134
780 IFB(F)=10THENPRINT"Q";	:rem	177
790 NEXT:PRINT5-I:NEXT	:rem	202
800 PRINTSPC(5)"{GIU'}{ 3 SPAZI}ABCDE{ 2 GIU'}"	:rem	87
810 RETURN	:rem	121
820 REM	:rem	126
830 OM=-1:WM=-1	:rem	75
840 FORSF=0TO41	:rem	145
850 P=B(SF(SF,0))+B(SF(SF,1))+B(SF(SF,2))+B(SF(SF,3))+B(SF(SF,4))	:rem	26
860 IFP=0ORP=10ORP=20ORP=30THENS0(SF)=10:GOTO960	:rem	5
870 IFP<>40THEN910	:rem	37
880 GOSUB970	:rem	192
890 GOSUB1010	:rem	227
900 WM=1:SF=41:GOTO960	:rem	50
910 IFP>20ORP<30ORP=11ORP=12THENS0(SF)=0:GOTO960	:rem	212
920 SO(SF)=1	:rem	149
930 IFP=4ANDOM=-1THENGOSUB970	:rem	69
940 IFP=14THENGOSUB1010	:rem	143
950 IFP=5THENSF=41	:rem	89
960 NEXT:RETURN	:rem	248
970 REM DETERMINA LE MOSSE OBBLIGATE (OM)	:rem	171
980 FORI=0TO4	:rem	21
990 IFB(SF(SF, I))=0THENOM=SF(SF, I)	:rem	41
1000 NEXT:RETURN	:rem	26
1010 REM	:rem	166

1020	FORI=0TO4	:rem	55
1030	IFB(SF(SF,I))<>10THEN1070	:rem	176
1040	FORJ=0TOCC-1	:rem	234
1050	IFSF(SF,I)=CC(J,0)THENC(J,1)=1	:rem	34
1060	NEXT	:rem	6
1070	NEXT	:rem	7
1080	RETURN	:rem	169
1090	REM	:rem	174
1100	NC=0:NP=0:NF=0	:rem	64
1110	FORI=0TO15	:rem	105
1120	DF(I,8)=-1	:rem	231
1130	DF(I,7)=0	:rem	185
1140	NEXT	:rem	5
1150	FORDF=0TO107	:rem	224
1160	O1=SO(D(DF,0)):O2=SO(D(DF,1))	:rem	90
1170	IFO1<>O2ORO1=0ORO2=0THEN1290	:rem	235
1180	GOSUB610	:rem	225
1190	GOSUB70	:rem	178
1200	IFP=4ORP=14THENDF(NF,7)=1:GOTO1230	:rem	78
1210	IFO1=1ORP<=DF(NF,8)THEN1290	:rem	121
1220	DF(NF,7)=10	:rem	53
1230	DF(NF,8)=P	:rem	38
1240	NF=0	:rem	200
1250	FORI=0TO15	:rem	110
1260	IFDF(I,8)>=DF(NF,8)THEN1280	:rem	40
1270	IFDF(I,7)=0ORDF(I,7)=10THENN=1	:rem	68
1280	NEXT	:rem	10
1290	NEXT	:rem	11
1300	RETURN	:rem	164
1310	REM VALUTA LA MOSSA STRATEGICA	:rem	109
1320	IFWM=1THEN1470	:rem	98
1330	MC=0:N2=0	:rem	235
1340	FORNF=0TO15	:rem	185
1350	IFDF(NF,7)=0THEN1410	:rem	140
1360	GOSUB70	:rem	177
1370	IFDF(NF,7)=1THEN1400	:rem	142
1380	IFP=MCTHENN2=N2+1	:rem	64
1390	IFP>MCTHENMC=P:N2=1	:rem	238
1400	IFP=4THENN=15	:rem	123
1410	NEXT	:rem	5
1420	IFPP=-1THEN1470	:rem	140
1430	IFP=4THENRETURN	:rem	39
1440	IFMC<MPTHENRETURN	:rem	208
1450	IFMC=MPANDN2<N1THENRETURN	:rem	224
1460	IFMC=MPANDN2=N1ANDRND(0)<.5THENRETURN	:rem	185
1470	MP=MC:N1=N2	:rem	172
1480	PP=CP:PT=CT	:rem	239
1490	RETURN	:rem	174
1500	REM MOSSA DEL GIOCATORE	:rem	159
1510	IFPC<5THEN1550	:rem	84
1520	PRINT"CHE PEDINA":PRINT"VUOI MUOVERE";	:rem	198

1530	GOSUB1630:PT=X	:rem	134
1540	IFB(PT)<>1THENPRINT"IMPOSSIBILE!":GOTO1520	:rem	215
1550	PRINT"DOVE VUOI METTERE":PRINT"LA TUA PEDINA";		
		:rem	145
1560	GOSUB1630:PP=X	:rem	133
1570	IFB(PP)<>0THENPRINT"IMPOSSIBILE!":GOTO1550	:rem	216
1580	IFPC=5THENB(PT)=0	:rem	53
1590	IFPC<5THENPC=PC+1	:rem	80
1600	B(PP)=1	:rem	104
1610	PRINT"{CLR}{ 2 GIU' }":GOSUB720	:rem	161
1620	RETURN	:rem	169
1630	REM INPUT	:rem	62
1640	INPUTX\$:IFLEN(X\$)<>2THEN1690	:rem	247
1650	L\$=LEFT\$(X\$,1):D\$=RIGHT\$(X\$,1)	:rem	157
1660	IFL\$<"A"ORL\$>"E"ORD\$<"1"ORD\$>"5"THEN1690	:rem	222
1670	X=ASC(L\$)-5*VAL(D\$)-40	:rem	180
1680	RETURN	:rem	175
1690	PRINT"INPUT ERRATO! RIPROVA":GOTO1640	:rem	128
1700	REM MOSSA DEL COMPUTER	:rem	115
1710	PRINT"{CLR}{GIU' }":IFCC<5THENPT=CC:CC=CC+1:GOTO1770	:rem	181
1720	PRINTSPC(5)"PRENDO{ 2 SPAZI}"CHR\$(65+PT-INT(PT/5)*5);5-INT(PT/5)	:rem	12
1730	B(PT)=0	:rem	111
1740	FORCI=0TO4	:rem	131
1750	IFCC(CI,0)=PTTHENPT=CI:CI=4	:rem	207
1760	NEXT	:rem	13
1770	PRINTSPC(4)"MUOVO IN{ 2 SPAZI}"CHR\$(65+PP-INT(PP/5)*5);5-INT(PP/5):PRINT	:rem	48
1780	B(PP)=10:CC(PT,0)=PP	:rem	143
1790	GOSUB720	:rem	234
1800	RETURN	:rem	169
1810	REM PARTE PRINCIPALE	:rem	17
1820	PRINT"{CLR}{BLK}{GIU' }":PRINTSPC(6)"ATTENDERE{ 2 GIU' }"	:rem	6
1830	GOSUB200	:rem	222
1840	GOSUB460	:rem	231
1850	CC=0:PC=0	:rem	251
1860	PP=12:GOTO2070	:rem	75
1870	FORI=0TO4:CC(I,1)=0:NEXT	:rem	225
1880	GOSUB820	:rem	235
1890	IFP=5THENPRINT"{GIU' }FINE.":END	:rem	149
1900	IFWM=1THEN1920	:rem	102
1910	GOSUB1090	:rem	21
1920	MP=0:N1=0:PP=-1	:rem	113
1930	BF=0:EF=24	:rem	42
1940	IFOM<>-1THENBF=OM:EF=OM	:rem	255
1950	FORCP=BFTOEF	:rem	60
1960	IFB(CP)<>0THEN2060	:rem	38
1970	B(CP)=10	:rem	149
1980	IFCC<5THENGOSUB1310:GOTO2050	:rem	6

1990 FORCI=0TO4	:rem	138
2000 IFCC(CI,1) THEN2040	:rem	6
2010 CT=CC(CI,0):B(CT)=0	:rem	39
2020 GOSUB1310	:rem	9
2030 B(CT)=10	:rem	141
2040 NEXT	:rem	5
2050 B(CP)=0	:rem	90
2060 NEXT	:rem	7
2070 GOSUB1700	:rem	17
2080 IFWM=1 THENP=5:GOTO1890	:rem	161
2090 GOSUB1500	:rem	17
2100 PRINTSPC(6)" { 2 GIU' }ATTENDERE"	:rem	191
2110 GOTO1870	:rem	205

Cinodromo

VIC-20

Questo programma per VIC 20 inespanso usa alcuni accorgimenti tecnici, quale la ridefinizione dei caratteri dello schermo, che lo rendono particolarmente gradevole graficamente.

Scopo del gioco è fare percorrere ad un levriero 200 metri di corsa il più velocemente possibile. Si può giocare da soli o in due persone. Il movimento del levriero è comandato dalla pressione di due tasti a

scelta del giocatore. Ovviamente il "repeat" su tasti come la barra spaziatrice viene disinserito.

Il programma è diviso in due parti, e come al solito la seconda va salvata sulla cassetta di seguito alla prima. Una volta caricata la prima parte nella memoria del tuo VIC, mandala in esecuzione con RUN ed essa provvederà automaticamente al caricamento della seconda.

programma 1

5	POKE52,28:POKE56,28:CLR:PRINT"{CLR}":PRINT"{HOME}		
	{ 3 GIU' }CARICAMENTO DA ROM"	:rem	110
10	FORI=7168TO7679:PRINT"{HOME}";I:POKEI,PEEK(I+25600):NEXT	:rem	4
11	PRINT"{HOME}{ 3 GIU' }CARICAMENTO IN RAM"	:rem	22
15	FORI=7168TO7383:PRINT"{HOME}";I:READA:POKEI,A:NEXT	:rem	242
		:rem	216
20	DATA0,0,0,1,3,3,23,255	:rem	185
25	DATA0,0,255,255,255,255,207,128	:rem	131
30	DATA8,12,254,255,224,204,67,32	:rem	61
35	DATA7,0,0,0,0,0,0,0	:rem	

```

40 DATA32,32,16,0,0,0,0,0      :rem 211
45 DATA0,0,0,1,3,5,24,1        :rem 119
50 DATA0,0,254,255,255,247,227,131 :rem 179
55 DATA8,12,254,255,224,192,128,0 :rem 137
60 DATA131,79,2,2,1,0,0,0      :rem 222
65 DATA0,0,0,3,5,8,48,0        :rem 133
70 DATA0,120,255,255,255,255,127,57 :rem 238
75 DATA8,12,254,255,224,192,128,0 :rem 139
80 DATA25,126,13,4,0,0,0,0     :rem 14
85 DATA128,64,192,0,0,0,0,0    :rem 76
90 DATA0,0,195,63,7,7,7,5      :rem 249
95 DATA0,7,255,255,255,223,207,192 :rem 195
100 DATA8,252,254,255,192,128,224,112 :rem 74
110 DATA13,12,12,24,16,8,0,0   :rem 111
120 DATA192,192,128,64,0,0,0,0 :rem 222
130 DATA46,64,,,,,,,,          :rem 182
135 DATA255,16,16,16,16,16,16,255 :rem 137
140 DATA60,66,129,165,129,153,66,60 :rem 244
145 DATA96,112,120,124,64,64,64,64 :rem 189
146 DATA0,0,0,31,31,55,39,39   :rem 131
147 DATA0,0,0,248,248,236,228,228 :rem 137
148 DATA63,31,7,7,3,1,1,15     :rem 33
149 DATA252,248,224,224,192,128,128,240 :rem 187
200 PRINT"{CLR}":POKE36879,27:PRINT"{BLK}" :rem 160
210 PRINT"{ 5 SPAZI}OM[<@>]R{ 2 F}R[<@>]NP" :rem 22
220 PRINT"{ 4 SPAZI}[<M>]{ 4 SPAZI}IU{ 4 SPAZI}[<G>]" :rem 79
230 PRINT"{ 4 SPAZI}Y O[<G>]{ 4 SPAZI}[<M>]P T" :rem 254
240 PRINT"{ 4 SPAZI}HN T{CYN}Q{ 2 SPAZI}Q{BLK}Y MG" :rem 15
250 PRINT"{ 7 SPAZI}G{ 4 SPAZI}H" :rem 247
260 PRINT"{ 7 SPAZI}B {BLK}[<D>][<F>]{BLK} -" :rem 143
270 PRINT"{ 7 SPAZI}HU{BLK}[<C>][<V>]{BLK}IG" :rem 49
280 PRINT"{ 7 SPAZI}Y- [<G>]-T" :rem 119
290 PRINT"{ 7 SPAZI}[<M>]J[<C>][<V>]K[<G>]" :rem 199
300 PRINT"{ 8 SPAZI}[<G>]{ 2 SPAZI}[<M>]" :rem 176
310 PRINT"{GIU'}{ 6 DES}{RED}C{CYN}I{PUR}N{BLU}O{GRN}D
{BLK}R{YEL}O{PUR}M{RED}O{BLU}" :rem 103
311 PRINT"{ 2 SU}U{ 20 *}I" :rem 38
312 PRINT"{SU}-CARICAMENTO IN CORSO-" :rem 245
313 PRINT"{SU}J{ 20 *}K{ 3 SU}{WHT}" :rem 70
314 POKE631,131:POKE198,1 :rem 138

```

programma 2

```

0 KIS="1900" :rem 51
1 POKE36869,242:POKE36879,108:PRINT"{WHT}{CLR}" :rem 22
2 PRINT"[RVS]{ 7 DES}CINODROMO{OFF}":PRINT"{YEL}
[< 22 Y>]" :rem 255
3 PRINT"{HOME}{ 7 GIU'}{ 8 DES}BY{ 2 SPAZI}V.F":PRINT"
{RVS}{CYN}{ 22 SPAZI}{OFF}":FOR T=1 TO 5000:NEXT :rem 109

```

```

10 INPUT"{CLR}{WHT}{ 3 DES}NUMERO GIOCATORI{HOME}
   { 2 GIU'"};A :rem 76
15 IFA<1THENPRINT"{CLR}FINE":GOTO20001 :rem 190
20 IFA>2THENPRINT"NON PIU'DI2":FORT=1TO1000:NEXT:GOTO10
   :rem 245
21 DIMA$(A):DIMDEC$(A):DIMTE(A):B=1 :rem 243
25 POKE36879,108:POKE36869,242:PRINT"{CLR}{WHT}DIMMI IL T
   UO{ 2 SPAZI}NOME":PRINT"GIOCATORE N.";B :rem 153
30 INPUT"NOME GIOCATORE{HOME}{ 3 GIU'"};A$(B):IFLEN(A$(B)
   )>3THENPRINT"{CLR}PER FAVORE 3 INIZIALI":GOTO30
   :rem 125
35 PRINT"OK {YEL}";A$(B){CYN} QUALE CANE{WHT} ":PRINT"P
   REFERISCI?" :rem 81
40 PRINT"{RVS}PLUTO{OFF} 1[<V>]":PRINT"{RVS}BUK{OFF} 2
   [<V>]":PRINT"{RVS}LILLI{OFF} 3[<V>]" :rem 115
45 GETQ$:IFQ$=""THEN45 :rem 21
50 IFQ$="1"THENB$="PLUTO":COL=0 :rem 202
51 IFQ$="2"THENB$="BUK":COL=2 :rem 28
52 IFQ$="3"THENB$="LILLI":COL=6 :rem 182
55 PRINT"OK ";A$(B);" FAI ":PRINT"{DES}CORRERE ";B$
   :rem 178
60 FORT=1TO500:NEXTT:PRINT"PREMI SHIFT PER IL VIA":S1=368
   77:WAIT653,1:TIS="000000" :rem 134
61 POKES1,220:F0R1=15TO0STEP-.1:POKE36878,L:NEXT:POKE3687
   8,15:POKES1,0:PRINT"{CLR}":POKE36879,219 :rem 35
62 POKE36869,255:POKE650,127 :rem 57
66 PRINT"{HOME}{RED}UUUUUUUUUUUUUUUUUUUUUUUUUUUUUU":PRINT"{RED}
   {SU}UUUUUUUUUUUUUUUUUUUUUUUUUUUUUU" :rem 192
67 PRINT"{SU}{RED}UUUUUUUUUUUUUUUUUUUUUUUUUUUUUU":PRINT"{RED}{SU}
   UUUUUUUUUUUUUUUUUUUUUUUUUUUUUUU" :rem 63
68 PRINT"{SU}{BLK}TTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTT":PRINT"{SU}{BLK}
   TTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTTT" :rem 252
69 PRINT"{GRN}V {BLK}V {BLU}V {RED}V {PUR}V {CYN}V {WHT}V
   {YEL}V {GRN}V{BLK} V {BLU}V ":PRINT"{ 4 GIU'":PRINT"
   {GRN}{ 22 SPAZI}" :rem 254
77 POKE8047,23:POKE8048,24:POKE8069,25:POKE8070,26:POKE38
   767,0 :rem 67
78 POKE38767+1,0:POKE38767+22,0:POKE38767+23,0 :rem 44
80 A1=38620:A2=38620+22 :rem 150
85 POKEA1,COL:POKEA1+1,COL:POKEA1+2,COL:POKEA2,COL:POKEA2
   +1,COL:POKEA2+2,COL :rem 6
90 TIS="000000":PRINT"{ 4 GIU'}" :rem 26
91 PL=8009 :rem 20
92 FORGI=1TO44 :rem 91
95 POKE7900,0:POKE7901,1:POKE7902,2:POKE7922,3:POKE7923,3
   2:POKE7924,4 :rem 3
100 POKE198,0:WAIT198,1:POKES1,240:POKES1,0:POKEPL,32:POK
   EPL-1,22 :rem 173
105 POKE7900,5:POKE7901,6:POKE7902,7:POKE7922,32:POKE7923
   ,8:POKE7924,32 :rem 112
110 POKE198,0:WAIT198,1:POKES1,220:POKES1,0:POKEPL,32:POK

```

```

EPL-1,22:PL=PL-1 :rem 185
111 IFPL<=7988THENPL=8009 :rem 240
120 POKE7900,9:POKE7901,10:POKE7902,11:POKE7922,32:POKE79
23,12:POKE7924,13 :rem 241
130 POKE198,0:WAIT198,1:POKES1,210:POKES1,0:POKEPL,32:POK
EPL-1,22:PL=PL-1 :rem 186
131 IFPL<=7988THENPL=8009 :rem 242
135 POKE7900,14:POKE7901,15:POKE7902,16:POKE7922,17:POKE7
923,18:POKE7924,19 :rem 60
140 POKE198,0:WAIT198,1:POKES1,240:POKES1,0:POKEPL,32:POK
EPL-1,22 :rem 177
141 POKE7988,32:IFPL<=7988THENPL=8009 :rem 205
142 NEXTGI:TE(B)=INT(TI/.6) :rem 117
5000 FORT=1TO500:NEXT:DEC$(B)=STR$(TE(B)) :rem 179
10000 FORY=38TO160:POKE36865,Y:NEXTY :rem 223
10010 POKE36869,240:PRINT"{CLR}":PRINT"{BLU}BENE ";A$(B):
PRINT"TUO TEMPO:" :rem 202
10011 PRINTLEFT$(DEC$(B),3);:PRINT"{DES}SEC.":;PRINT"
{DES}& ";:PRINTRIGHT$(DEC$(B),2);:PRINT" CEN."
:rem 23
10020 FORY=160TO38STEP-1:POKE36865,Y:NEXTY:B=B+1:IFB>ATHE
N20000 :rem 125
10025 FORT=1TO2000:NEXT:PRINT"{HOME}":GOTO25 :rem 119
10030 FORY=12TO84:POKE36864,Y:FORT=1TO10:NEXTT,Y:rem 21
10040 PRINT"{CLR}":PRINT"ORA TOCCA AL GIOCATORE":PRINT"
{ 2 SPAZI}NUMERO";B :rem 56
10050 FORY=84TO12STEP-1:POKE36864,Y:FORT=1TO30:NEXTT,Y:ST
OP :rem 51
20000 FORT=1TO2000:NEXT:ONAGOTO30000,30100 :rem 221
20001 POKE646,INT(RND(1)*6)+1:PRINT"{CLR}GIOCHI ANCORA?<S
/N>":GETA$:IFA$=""THEN20001 :rem 42
20002 IFA$<>"S"ANDA$<>"N"THEN20001 :rem 95
20003 IFA$="S"THENRUN :rem 225
20004 SYS64802 :rem 249
30000 IFVAL(DEC$(1))<VAL(KI$)THENPRINT"{CLR}HAI BATTUTO I
L ":PRINT"{RVS}{BLK}R{RED}E{CYN}C{PUR}O{GRN}R{YEL}D
{BLU}!!!":GOTO50000 :rem 214
30010 IFVAL(DEC$(1))=VAL(KI$)THENPRINT"{CLR}{RVS}BRAVO HA
I EGUALIATO{OFF}":PRINT"IL {BLK}R{RED}E{CYN}C{GRN}O
{BLU}R{YEL}D{BLU}":GOTO50000 :rem 124
30020 IFVAL(DEC$(1))>VAL(KI$)THENPRINT"RI PROVACI!!":FORT=
1TO5000:NEXT:GOTO20001 :rem 111
30100 IFVAL(DEC$(1))<VAL(DEC$(2))THENPRINT"{CLR}VINCE ";A
$(1) :rem 54
30110 IFDEC$(2)<DEC$(1)THENPRINT"{CLR}VINCE ";A$(2)
:rem 208
50000 H$="21721322322723423022723423022322721722321721322
3227234230227234230223" :rem 142
50001 FORY=0TO22:POKE36876,VAL(MID$(H$,Y*3+1,3)):FORYY=15
TO0STEP-.5:POKE36878,YY:NEXT :rem 190
50002 NEXT:POKE36876,0:GOTO20001 :rem 111

```

Questa volta, affezionati lettori di Papersoft, si tratta di un gioco spaziale, che utilizza lo scroll naturale del computer, ovvero la parola chiave PRINT. Voi guidate l'astronave, e dovete schivare per mezzo dei tasti <, > e <.> gli asteroidi scagliati contro da una delle solite tempeste spaziali (che capitano sempre quelle poche volte che capita di guidare un astronave).

Dovrete scegliere la "densità asteroidi" (più è alta e più è difficile), e il livello di difficoltà (stesso discorso).

Naturale che a compimento della missione sarete ricompensati dal Ministro del Pianeta di appartenenza, mentre in caso di fallimento subirete un processo non troppo regolare (come al solito!).

```
100 RANDOMIZE
110 CALL CLEAR
120 FOR I=5 TO 8
130 CALL COLOR(I,16,1)
140 NEXT I
150 CALL CLEAR
160 PRINT "  A S T R O S T O R M !!!"
170 PRINT
180 PRINT
190 PRINT "MUOVI LA NAVE CON I TASTI < & >"
200 FOR I=1 TO 9
210 PRINT
220 NEXT I
230 FOR I=14 TO 3 STEP -1
240 FOR J=1 TO 20
250 NEXT J
260 CALL SCREEN(I)
270 NEXT I
280 FOR I=1 TO 400
290 NEXT I
300 S=3
310 PT=0
320 RSHIP=10
330 CSHIP=1
340 SHIP=62
```

```

350 COL=2
360 RLSHIP=RSHIP
370 CLSHIP=CSHIP
380 CALL CLEAR
390 PRINT "DENSITA' ASTEROIDI (1-10)";
400 INPUT X
410 IF (X>10)+(X<1) THEN 380
420 PRINT
430 PRINT "LIVELLO DI DIFFICOLTA (1-10)"
440 INPUT DCULT
450 IF (DCULT>10)+(DCULT<1) THEN 380
460 CALL CLEAR
470 CALL SCREEN(S)
480 CALL COLOR(2,16,1)
490 REM
500 FOR LOOP=1 TO 1000
510 FOR I=1 TO INT(RND*X)+1
520 COL=INT(RND*30)+2
530 CALL HCHAR(23,COL,42)
540 NEXT I
550 PRINT
560 CALL GCHAR(RSHIP+DCULT,CSHIP,LOC)
570 CALL HCHAR(RLSHIP+DCULT-1,CLSHIP,32)
580 IF LOC=42 THEN 1190
590 CALL SCREEN(S)
600 CALL HCHAR(RSHIP+DCULT,CSHIP,SHIP)
610 CLSHIP=CSHIP
620 RLSHIP=RSHIP
630 CALL KEY(3,A,STATUS)
640 IF A<>ASC(".") THEN 730
650 CALL SOUND(-700,-5,3)
660 PT=PT+5*DCULT*X
670 CSHIP=CSHIP+1
680 IF CSHIP<>32 THEN 720
690 CSHIP=1
700 S=S+1
710 IF S=15 THEN 1030
720 SHIP=62
730 IF A<>ASC(",") THEN 820
740 CALL SOUND(-700,-6,4)
750 SHIP=60
760 PT=PT-8*DCULT*X
770 CSHIP=CSHIP-1
780 IF CSHIP>0 THEN 810

```

```

790 CSHIP=32
800 S=S-SGN(S-3)
810 SHIP=60
820 REM
830 NEXT LOOP
840 CALL CLEAR
850 PRINT " NON HAI COMPLETATO LA TUA"
860 PRINT
870 PRINT "MISSIONE. IL GOVERNO DI"
880 PRINT
890 PRINT "SOLARIAN TI HA PRESO"
900 PRINT
910 PRINT PT;"IBM PC DALLA TUA CASSA"
920 PRINT
930 PRINT
940 PRINT
950 PRINT " PERO TI HA ANCHE DATO"
960 PRINT
970 PRINT "UNA MISSIONE DA COMPIERE!!!"
980 PRINT
990 PRINT
1000 INPUT "CI PROVI (S O N)?:G$
1010 IF G$="S" THEN 300
1020 GOTO 1500
1030 REM
1040 CALL CLEAR
1050 PRINT " HAI COMPLETATO LA TUA "
1060 PRINT
1070 PRINT "MISSIONE. IL MINISTO DI"
1080 PRINT
1090 PRINT "SOLARIAN TI HA DATO ";PT
1100 PRINT
1110 PRINT "ZX-80 PER IL TUO SERVIZIO"
1120 PRINT
1130 PRINT "E SI AUGURA CHE TU COMPIA IL"
1140 PRINT
1150 PRINT "PROSSIMO COMPITO AFFIDATOTI!"
1160 PRINT
1170 PRINT
1180 GOTO 980
1190 CALL SCREEN(12)
1200 FOR I=1 TO 50
1210 NEXT I
1220 CALL SCREEN(9)

```

```

1230 FOR VOL=24 TO 1 STEP -4
1240 CALL SOUND(200,-7,VOL)
1250 NEXT VOL
1260 FOR VOL=1 TO 24 STEP 4
1270 CALL SOUND(200,-7,VOL)
1280 NEXT VOL
1290 CALL SCREEN(12)
1300 FOR I=1 TO 10
1310 NEXT I
1320 CALL SCREEN(8)
1330 CALL CLEAR
1340 CALL SCREEN(6)
1350 PRINT " MALISSIMO! IL GOVERNO DI"
1360 PRINT " SOLARIAN TI HA SPEDITO UN"
1370 PRINT " CREDITO DI";PT;"DOLLARI "
1380 PRINT " AMERICANI."
1390 FOR I=1 TO 5
1400 PRINT
1410 NEXT I
1420 PRINT " PREMI -P- PER RITENTARE"
1430 PRINT " OPPURE -S- PER FINIRE."
1440 FOR I=1 TO 5
1450 PRINT
1460 NEXT I
1470 CALL KEY(3,K,ST)
1480 IF ST=0 THEN 1470
1490 IF (K=ASC("P")) THEN 300
1500 IF (K=ASC("S")) THEN 1510 ELSE 1470
1510 CALL CLEAR
1520 PRINT TAB(8);"ARRIVEDERCI!":::::::::::::
1530 END

```

Caccia ai rombi

BASIC

Questa volta vi dovrete sfogare con qualche centinaio di piccoli oggetti a forma di rombo che somigliano tanto a dei grossi diamanti....

Questi preziosi oggetti sono sparpagliati al suolo, a vostra disposizione.

Per arricchirvi raccogliendoli non dovrete fare altro che utilizzare i

tasti con le frecce per muoversi sopra di essi, facendo però attenzio-

ne ai non pochi imprevisti che ostacoleranno la vostra "vendemmia".

```
100 CALL CLEAR
110 PRINT TAB(6);"*****"
120 PRINT TAB(6);"* CACCIA AI ROMBI *"
130 PRINT TAB(6);"*****"
140 PRINT ::"USA I TASTI FRECCIATI PER"
150 PRINT : "MUOVERTI E PRENDERE TUTTI I"
160 PRINT : "ROMBI CHE STANNO"
170 PRINT : "SUL SUOLO."
180 PRINT ::"ATTENZIONE AL CACCIATORE!":::
190 CALL CHAR(128,"0000183C3C18")
200 CALL CHAR(96,"001C1C083E081414")
210 CALL COLOR(9,7,1)
220 CALL CHAR(104,"1C1C083E08141414")
230 PRINT "PREMI UN TASTO PER PARTIRE.";
240 CALL KEY(0,K,S)
250 IF S<1 THEN 240
260 T=0
270 CALL CLEAR
280 X=1
290 Y=3
300 RANDOMIZE
310 C=INT(15*RND+2)
320 IF C=4 THEN 310
330 CALL COLOR(13,C,1)
340 CALL HCHAR(1,1,128,768)
350 A=INT(24*RND+1)
360 B=INT(25*RND+5)
370 PG=128
380 CALL SOUND(150,1497,2)
390 CALL HCHAR(X,Y,96)
400 CALL SOUND(150,110,2)
410 CALL HCHAR(A,B,104)
420 CALL KEY(1,K,S)
430 IF (K+1<1)+(K>5) THEN 680
440 ON K+1 GOTO 450,680,480,510,680,540
450 DX=1
460 DY=0
470 GOTO 560
480 DX=0
490 DY=-1
500 GOTO 560
```

```

510 DX=0
520 DY=1
530 GOTO 560
540 DX=-1
550 DY=0
560 CALL HCHAR(X,Y,32)
570 X=X+DX
580 Y=Y+DY
590 X=INT(24*((X-1)/24-INT((X-1)/24)))+1
600 Y=INT(32*((Y-1)/32-INT((Y-1)/32)))+1
610 CALL GCHAR(X,Y,G)
620 IF G=104 THEN 830
630 IF G=32 THEN 670
640 CALL SOUND(-100,1497,2)
650 T=T+1
660 IF T=767 THEN 870
670 CALL HCHAR(X,Y,96)
680 CALL HCHAR(A,B,PG)
690 A=A-SGN(A-X)
700 IF A>0 THEN 720
710 A=1
720 IF A<25 THEN 740
730 A=24
740 B=B-SGN(B-Y)
750 IF B>0 THEN 770
760 B=1
770 IF B<32 THEN 790
780 B=32
790 CALL GCHAR(A,B,PG)
800 IF PG=96 THEN 830
810 CALL HCHAR(A,B,104)
820 GOTO 420
830 CALL SOUND(300,-5,2)
840 PRINT "IL CACCIATORE TI HA PRESO!!"
850 PRINT : "HAI PRESO";T;"ROMBI."
860 GOTO 890
870 PRINT "CONGRATULAZIONI!!"
880 PRINT "HAI PRESO TUTTI I ROMBI!!"
890 PRINT :: "RITENTI? (S/N)";
900 CALL KEY(0,K,S)
910 IF K=83 THEN 260
920 IF K<>78 THEN 900
930 PRINT :::::
940 END

```

Ma qui è pieno di bug!

16/48K



Eccovi imprigionati in un'area RAM del vostro Spectrum, stretti d'assedio dai bug che voi stessi avete introdotto nel corso del vostro ultimo programma BASIC! Le vostre sembianze sono quelle di un raggio di luce pura: dovete dunque muovervi senza toccare nulla sullo schermo (men che meno i ferocissimi

mi bug!), in attesa che una bomba ad orologeria posta al centro del video esploda, purificando quell'area da tutti gli ospiti indesiderati. Usate i tasti "Q" e "W" per muovervi a sinistra e destra e "L" e "P" per spostarvi in basso e in alto. Il programma gira anche su Spectrum 16K.

```
1 REM -----
2 REM MA·QUI·E'·PIENO·DI·BUG!
3 REM -----
10 LET keys=0
20 LET BUG=1
30 LET C=1
40 BORDER 0: PAPER 0: INK 5: CLS
70 FLASH 0: CLS
90 POKE 23658,255
100 PRINT AT 10,3;"PREGO, ATTENDI UN ATTIMO..."
130 REM UDG1
140 FOR G=USR "C" TO USR "F"+7: READ D: POKE G,D: NEXT G
150 DATA 129,247,236,89,60,63,120,112
152 DATA 129,239,55,154,60,252,30,14
154 DATA 96,40,31,15,30,12,116,240
156 DATA 6,12,248,240,123,54,46,28
160 REM UDG2
170 FOR G=USR "G" TO USR "J"+7: READ D: POKE G,D: NEXT G
180 DATA 121,247,236,153,188,63,127,113
182 DATA 158,239,55,153,61,252,254,142
184 DATA 102,63,31,15,222,108,116,56
186 DATA 102,252,248,240,120,48,46,15
220 REM UDG3
230 FOR G=USR "N" TO USR "N"+7: READ D: POKE G,D: NEXT G
240 DATA 0,0,129,195,195,195,129,0
245 REM UDG4
250 FOR G=USR "O" TO USR "T"+7: READ D: POKE G,D: NEXT G
260 DATA 199,255,91,61,35,117,121,255
261 DATA 129,231,255,255,255,219,255,255
262 DATA 227,255,218,188,196,174,158,255
263 DATA 202,234,96,106,106,254,63,7
264 DATA 170,170,0,0,170,170,255,255
265 DATA 163,167,6,166,166,190,252,224
270 REM UDG5
271 FOR G=USR "U" TO USR "U"+7: READ D: POKE G,D: NEXT G
275 DATA 126,255,255,255,255,255,255,129
280 BORDER 0: PAPER 0: INK 5: CLS
281 LET D=1
290 FLASH 1: INK 6: LET A$="MA QUI E'": GO SUB 400: FLASH 0
```

```

750 LET J=H: LET H=0: LET H=H-(N=81)+(N=87): IF H=0 THEN LET H=J
760 LET J=V: LET V=0: LET V=V-(N=76)+(N=80): IF V=0 THEN LET V=J
762 LET H=H+(H=1)-(H=-1)
764 LET V=V+(V=1)-(V=-1)
770 LET X=X+H: LET Y=Y+V
771 LET T=INT (RND*10)+1: REM il grande BUG
772 IF T>7 THEN PRINT AT 9,25;"{3G8}";AT 10,25;"{3G8}": PRINT AT 9,
2; INK 6; BRIGHT 1;"OPQ";AT 10,2;"RST": BEEP .002,40: BEEP .002,
50: BEEP .002,60
773 IF T<3 THEN PRINT AT 9,2;"{3G8}";AT 10,2;"{3G8}": PRINT AT 9,25
; INK 6; BRIGHT 1;"OPQ";AT 10,25;"RST": BEEP .002,40: BEEP .002,
50: BEEP .002,60

775 NEXT Z
780 REM movimento dei BUGS
790 BEEP .001,10: BEEP .001,20: PRINT AT 14,M(1);"{2G8}"; INK 3;H$(B
UG);"{2G8}": PRINT AT 15,M(1);"{2G8}"; INK 3;L$(BUG);"{2G8}"
800 BEEP .001,-10: BEEP .001,-20: PRINT AT 5,M(2);"{2G8}"; INK 6;H$(
BUG);"{2G8}": PRINT AT 6,M(2);"{2G8}"; INK 6;L$(BUG);"{2G8}"
810 IF BUG=1 THEN LET BUG=2: GO TO 830
820 LET BUG=1
830 LET M(1)=M(1)+O: LET M(2)=M(2)+P
840 IF M(1)<10 THEN LET O=2
850 IF M(1)>17 THEN LET O=-2
860 IF M(2)<10 THEN LET P=2
870 IF M(2)>17 THEN LET P=-2
880 PRINT AT 10,16; INK 7; BRIGHT 1; INVERSE 1;"{2G8}"
885 PRINT AT 10,16; INK 7; BRIGHT 1; INVERSE 1;"0" AND TM<10);TM
890 PRINT AT 21,8; PAPER 2; INK 6;"PUNTI=";SC;"{7G8}";AT 21,20;"VITE
=";MEN
895 BEEP .002,40: LET L=INT (RND*12)+5: LET Q=INT (RND*29)+2
896 IF ATTR (L,Q)=71 THEN GO TO 895
897 PRINT AT L,Q; INK 4; FLASH 1;"{SG6}"
900 NEXT A: LET SC=SC+5: LET TM=TM-1: BEEP .01,-2: IF TM>0 THEN
GO TO 700
999 REM LA BOMBA ESPLODE
1000 DIM B$(3,4): LET B$(1)="{G4}{SG6}{SG7}{G7}"
1001 LET B$(2)="{SG3}{SG4}{SG8}{G6}"
1002 LET B$(3)="{SG2}{SG6}{G7}{G6}"
1005 PRINT AT 7,12; INK 5;"BONUS ";BO*10
1006 LET SC=SC+BO*10
1011 FOR G=1 TO 20: FOR J=1 TO 3
1020 LET F=INT (RND*3)+1: LET K=INT (RND*-5)-10
1030 BEEP .003,K+10: PRINT AT 8+J,15;B$(F)
1040 NEXT J: NEXT G
1050 PAUSE 150: CLS
1060 PRINT AT 2,15; INK 4;"OPQ";AT 3,15; INK 4;"RST"
1070 LET D=16
1080 PRINT : PRINT : LET A$="TI PRENDERO' LA PROSSIMA VOLTA !":
GO SUB 400
1090 PRINT : PRINT
1110 INK 6: LET A$="ATTENZIONE AL PROSSIMO LIVELLO !": GO SUB 400:
INK 5
1120 PAUSE 0: CLS : LET X=240: LET Y=88: LET H=-2: LET V=2: LET TM=BO
+5: GO TO 580
2000 LET X=240: LET Y=88: REM perdi una vita
2010 FOR J=40 TO -5 STEP -1
2020 BEEP .04,J: BORDER INT (RND*7)+1: NEXT J
2025 LET MEN=MEN-1
2030 BORDER 0: CLS
2040 FLASH 1: FOR G=0 TO 21
2045 LET X$="{G7}{SG7}{G7}{SG7}{G7}{SG7}{G7}{SG7}{G7}{SG7}{G7}{SG7}"

```

```

300 LET A$="PIENO": GO SUB 400
310 INVERSE 1: LET A$="DI BUG!!!": GO SUB 400: INVERSE 0: INK 5
320 LET A$="VUOI VEDERE LE ISTRUZIONI (S/N)?": GO SUB 400: PAUSE 0:
    LET D$=CHR$(PEEK 23560): IF D$="N" THEN GO TO 500
330 CLS : LET D=1: PRINT : LET A$="I S T R U Z I O N I": GO SUB 400
340 LET A$="Sei imprigionato in un'area RAM ": GO SUB 400
341 LET A$="dello SPECTRUM, infestata da": GO SUB 400
342 LET A$="terribili BUGS! Devi cercare di ": GO SUB 400
343 LET A$="muoverti sullo schermo senza": GO SUB 400
344 LET A$="toccarli. Il tuo aspetto e' ": GO SUB 400
356 LET A$="quello di un raggio di luce pura": GO SUB 400
357 LET A$="in perenne movimento....": GO SUB 400
359 FLASH 1: INK 7: BRIGHT 1: LET A$=" PREMI UN TASTO PER CONTINUAR
    E ": GO SUB 400: PAUSE 0: FLASH 0: CLS : LET D=1
360 LET A$="TASTI/CONTROLLO": GO SUB 400
361 INK 6: LET A$="Q-----SINISTRA": GO SUB 400

362 INK 6: LET A$="W-----DESTRA ": GO SUB 400
363 INK 6: LET A$="L-----BASSO ": GO SUB 400
364 INK 6: LET A$="P-----ALTO ": GO SUB 400
365 LET D=D-1: LET A$="
    ": GO SUB 400
    : LET A$="NON TOCCARE NULLA SULLO SCHERMO": GO SUB 400: LET A$="
    ...O MORIRAI !!!": LET D=D-1: GO SUB 400: INK 5: LET A$="C'e' un
    a bomba a orologeria nel": GO SUB 400
366 LET a$="mezzo dello schermo. Quando": GO SUB 400
367 LET a$="esplode uccide tutti i BUGS!!": GO SUB 400
370 FLASH 1: INK 7: BRIGHT 1: LET A$=" PREMI UN TASTO PER GIOCARE
    ": GO SUB 400: FLASH 0: PAUSE 0: GO TO 500
400 LET X=LEN (A$): LET X=INT (32-X)/2
410 FOR G=1 TO LEN A$: BEEP .004,6
420 PRINT AT D,X+G-1;A$(G TO G): NEXT G: LET D=D+2
430 RETURN
440 REM INIZIA IL GIOCO
500 INK 5: PAPER 0: BORDER 0: CLS
505 REM INIZIALIZZAZIONE VARIABILI
510 LET SC=0: LET MEN=3: LET TM=20: LET BO=TM
520 DIM H$(2,2): DIM L$(2,2)
530 LET H$(1)="CD": LET H$(2)="GH"
540 LET L$(1)="EF": LET L$(2)="IJ"
550 LET X=240: LET Y=88
560 LET H=-3: LET V=-3: DIM M(2): LET M(1)=16: LET M(2)=16
580 LET BUG=1
600 REM ALLESTIMENTO SCHERMO
610 PLOT 252,8: DRAW 0,167: PLOT 254,8: DRAW 0,166: PLOT 2,8: DRAW 9
    5,0: DRAW 26,30: DRAW 20,0: DRAW 26,-30: DRAW 84,0: DRAW 0,166:
    DRAW -84,0: DRAW -26,-30: DRAW -20,0: DRAW -26,30: DRAW -95,0:
    DRAW 0,-166
615 PRINT AT 21,0;"{32SG8}"
620 FOR G=-10 TO 7: FOR K=1 TO 5: BEEP .003,G+H: NEXT K: NEXT G
630 PRINT AT 9,15: INK 7: BRIGHT 1;"{G4}{2SG8}{SG7}"
640 PRINT AT 10,15: INK 7: BRIGHT 1;"{G5}"; INVERSE 1;"0" AND TM<10
    );TM; INVERSE 0;"{SG5}"
650 PRINT AT 11,15: INK 7: BRIGHT 1;"{G1}{2SG8}{G2}"
660 LET O=-2: LET P=2
700 REM LOOP PRINCIPALE
710 FOR A=1 TO 2: PLOT INK 4;X,Y: BEEP .002,0: BEEP .002,-4
711 FOR Z=1 TO 6: PLOT INK 4;X,Y: BEEP .002,0: BEEP .002,-4
720 LET HIT=POINT (X+H,Y+V): IF HIT=1 THEN GO TO 2000
725 LET HIT=POINT (X+H,Y+V-V): IF HIT=1 THEN GO TO 2000
726 LET HIT=POINT (X+H,Y+V-(V-1)): IF HIT=1 THEN GO TO 2000
730 REM LETTURA TASTIERA
740 LET N=PEEK (23560)

```

```

{G7}{SG7}{G7}{SG7}{G7}{SG7}{G7}{SG7}{G7}{SG7}{G7}{SG7}{G7}{SG7}
{G7}{SG7}{G7}{SG7}{G7}{SG7}"
2050 PRINT PAPER 2; INK 6; BRIGHT 1;X$: NEXT G
2060 FLASH 0: PRINT AT 9,4;"ACCIDENTI... SEI MORTO!!!"
2070 IF MEN>0 THEN PRINT AT 11,9;"VITE RESTANTI ";MEN
2080 IF MEN=0 THEN PRINT AT 11,6;"LA PARTITA E' FINITA": GO TO 2100
2090 PAUSE 0: LET H=-2: LET V=-2: CLS : GO TO 580
2100 REM RIPOSA IN PACE
2105 PAUSE 0
2110 CLS : PRINT : PRINT
2120 PRINT TAB 12;"{3G8}{SG8}"
2130 PRINT TAB 12;" R.I.P "
2140 FOR G=1 TO 6: PRINT TAB 12;"{3G8}{SG8}": NEXT G
2150 PRINT : PRINT
2160 PRINT INK 6;"{6G8}IL PUNTEGGIO E' : ";SC: PRINT : PRINT "{4G8}
{9G8}OPQ": PRINT "{3G8} {10G8}RST"
2180 PRINT : PRINT "{3G8}PREMI UN TASTO PER GIOCARE{3G8}": PAUSE 0:
GO TO 280

```

ERRATA CORRIGE

A causa di uno spiacevole errore, due linee DATA del quarto Papersoft n. 26 del 5 luglio 1985, presentano errori tali da impedirne la corretta esecuzione. Le linee corrette sono riportate qui di seguito.

```

80 DATA 75,111,122,205,131,122,201,33,30,123,17,37,0,25,61,
32,252,229,221,225
90 DATA 201,0,0,0,0,0,0,0,0,5081

```

Guida per l'input dei programmi sullo ZX Spectrum

Come è noto, lo Spectrum è provvisto di 2 serie di tasti grafici: una prima serie di 16 caratteri grafici predefiniti (i tasti numerici da 1 a 8 e gli stessi "shiftati") e una serie di caratteri definibili dall'utente nell'ambito di un programma (le lettere da A a U). In entrambi i casi, per ottenere i caratteri desiderati occorre entrare in modo grafico (cursore contrassegnato dalla lettera G lampeggiante) premendo contemporaneamente i tasti CAPS SHIFT e 9.

Nei nostri listati i caratteri grafici predefiniti sono indicati da una G e dal numero corrispondente al tasto che occorre digitare, il tutto racchiuso tra due parentesi graffe.

A esempio {G4} significa che occorre digitare il tasto 4, con il cursore in modo grafico.

Analogamente la codifica SG, seguita da un numero da 1 a 8, significa che occorre digitare il relativo tasto numerico premendo contemporaneamente il tasto CAPS SHIFT.

Ad esempio quando si trova la codifi-

ca {SG2}, occorre premere il tasto 2 contemporaneamente al tasto CAPS SHIFT, ovviamente con il cursore in modo grafico. In entrambi i casi precedenti, quando un simbolo grafico deve essere digitato più volte, i caratteri G o SG della codifica sono preceduti da un numero che specifica quante volte va premuto il tasto grafico indicato.

Così ad esempio {8G5} significa che il tasto grafico 5 va digitato 8 volte; analogamente {4SG1} significa che il tasto grafico 1, premuto insieme a CAPS SHIFT, deve essere battuto 4 volte. I caratteri grafici definibili (le lettere da A a U in modo grafico) hanno una codifica semplificata: la lettera corrispondente, sottolineata.

Quando in un listato viene presentata, ad esempio, una A sottolineata, occorre entrare in modo grafico (al solito premendo contemporaneamente i tasti CAPS-SHIFT e 9) e quindi digitare semplicemente il tasto che contrassegna la lettera A.

Quando leggete	Premete	Vedrete
{G1}		
{G2}		
{G3}		
{G4}		
{G5}		
{G6}		
{G7}		
{G8}		
{SG1}		
{SG2}		
{SG3}		
{SG4}		
{SG5}		
{SG6}		
{SG7}		
{SG8}		

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Quando leggete	Premete	Vedrete
<u>A</u>		
<u>B</u>		
<u>C</u>		
<u>D</u>		
<u>E</u>		
<u>F</u>		
<u>G</u>		
<u>H</u>		
<u>I</u>		
<u>J</u>		
<u>K</u>		
<u>L</u>		
<u>M</u>		
<u>N</u>		
<u>O</u>		
<u>P</u>		
<u>Q</u>		
<u>R</u>		
<u>S</u>		
<u>T</u>		
<u>U</u>		

Simbolo grafico definito nel programma in uso.

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

SCALDA IL JOYSTICK E GASATI CON



*ogni mese
in edicola*



**GRUPPO
EDITORIALE
JACKSON**

Milano-Londra-Madrid-San Francisco

IL VERO GIOCO COMINCIA ADESSO